Taylor Series Model

In[575]:=
```
S[x_] = Normal[Series[f[x], {x, 0, 4}]] (*Taylor Series expansion*)
```

Out[575]= $f[0] + x f'[0] + \frac{1}{2} x^2 f''[0] + \frac{1}{6} x^3 f^{(3)}[0] + \frac{1}{24} x^4 f^{(4)}[0]$

In[576]:= $S\left[\frac{0+h}{2}\right]$ (*Example*)

Out[576]= $f[0] + \frac{1}{2} h f'[0] + \frac{1}{8} h^2 f''[0] + \frac{1}{48} h^3 f^{(3)}[0] + \frac{1}{384} h^4 f^{(4)}[0]$

Definite Integral of Taylor Series

In[577]:=
```
Exact = Simplify[Integrate[S[x], {x, 0, h}]] (*Exact integral of S(x) on [a,b]*)
```

Out[577]= $h f[0] + \frac{1}{120} h^2 \left(60 f'[0] + h \left(20 f''[0] + h \left(5 f^{(3)}[0] + h f^{(4)}[0]\right)\right)\right)$

Midpoint Integration

In[578]:= $\text{Midpoint} = \text{Simplify}\left[h\, S\left[\frac{h}{2}\right]\right]$

Out[578]= $h f[0] + \frac{1}{384} h^2 \left(192 f'[0] + h \left(48 f''[0] + h \left(8 f^{(3)}[0] + h f^{(4)}[0]\right)\right)\right)$

In[579]:=
```
Expand[Simplify[Midpoint - Exact]]
```

Out[579]= $-\frac{1}{24} h^3 f''[0] - \frac{1}{48} h^4 f^{(3)}[0] - \frac{11 h^5 f^{(4)}[0]}{1920}$

Trapezoid Integration

In[580]:= $\text{Trap} = \text{Simplify}\left[h \frac{S[0] + S[h]}{2}\right]$ (*Series expansion of trapezoid rule about c*)

Out[580]= $h f[0] + \frac{1}{48} h^2 \left(24 f'[0] + h \left(12 f''[0] + h \left(4 f^{(3)}[0] + h f^{(4)}[0]\right)\right)\right)$

In[581]:=
```
Expand[Simplify[Trap - Exact]]
```

Out[581]= $\frac{1}{12} h^3 f''[0] + \frac{1}{24} h^4 f^{(3)}[0] + \frac{1}{80} h^5 f^{(4)}[0]$

In[582]:=
```
(*How does this compare to Midpoint??*)
```

Simpson Rule

In[583]:=
```
Simp = FullSimplify[h/6 (S[0] + 4 S[h/2] + S[h])]
```

Out[583]= $h\,f[0] + \frac{1}{576} h^2 \left(288\,f'[0] + h\left(96\,f''[0] + h\left(24\,f^{(3)}[0] + 5\,h\,f^{(4)}[0]\right)\right)\right)$

In[584]:=
```
Expand[Simplify[Simp - Exact]]
```

Out[584]= $\frac{h^5\,f^{(4)}[0]}{2880}$

Conditioning of Interpolatory Quadrature Schemes

In[585]:=
```
nQ = 10;  (* a=1, b=nQ, ... Interval width: b-a = nQ-1 *)
```

In[586]:=
```
UnitVector[nQ, 2]
```

Out[586]= {0, 1, 0, 0, 0, 0, 0, 0, 0, 0}

In[587]:=
```
Clear[poly]
```

In[588]:=
```
poly[n_, k_] := InterpolatingPolynomial[UnitVector[n, k], x]
```

In[589]:=
```
Plot[Table[poly[nQ, k], {k, 1, nQ}], {x, 1, nQ}, PlotRange → {-1, 1}]
```

Out[589]=



In[590]:=
```
quadWeights = N[Integrate[Table[poly[nQ, k], {k, 1, nQ}], {x, 1, nQ}]]
```

Out[590]= {0.286975, 1.58113, 0.108482, 1.94304,
0.580379, 0.580379, 1.94304, 0.108482, 1.58113, 0.286975}

In[591]:=
```
ListPlot[Sort[quadWeights]]
```

Out[591]=



In[592]:=
```
nQ - 1
```

Out[592]= 9

In[593]:=
```
Sum[quadWeights[[i]], {i, nQ}]
```

Out[593]= 9.

In[594]:=
```
Norm[quadWeights, 1]
```

Out[594]= 9.

In[595]:=
```
ConditionNumber = Norm[quadWeights, 1] / (nQ - 1)
```

Out[595]= 1.

In[596]:=
```
quadWeights
```

Out[596]=
```
{0.286975, 1.58113, 0.108482, 1.94304,
 0.580379, 0.580379, 1.94304, 0.108482, 1.58113, 0.286975}
```

Finite Differences

In[597]:=
```
(*Taylor series of f about x=c: *)
S[x_] = Normal[Series[f[x], {x, c, 4}]] (*Longer Taylor Series expansion*)
```

Out[597]= $f[c] + (-c + x)\, f'[c] + \frac{1}{2} (-c + x)^2 f''[c] + \frac{1}{6} (-c + x)^3 f^{(3)}[c] + \frac{1}{24} (-c + x)^4 f^{(4)}[c]$

In[598]:=
```
S[c + h]
```

Out[598]= $f[c] + h\, f'[c] + \frac{1}{2} h^2 f''[c] + \frac{1}{6} h^3 f^{(3)}[c] + \frac{1}{24} h^4 f^{(4)}[c]$

In[599]:=
```
FwdDiff = Series[(S[c + h] - S[c])/h, {h, 0, 2}]
```

Out[599]= $f'[c] + \frac{1}{2} f''[c]\, h + \frac{1}{6} f^{(3)}[c]\, h^2 + O[h]^3$

In[600]:=
```
BwdDiff = Series[(S[c] - S[c - h])/h, {h, 0, 2}]
```

Out[600]= $f'[c] - \frac{1}{2} f''[c]\, h + \frac{1}{6} f^{(3)}[c]\, h^2 + O[h]^3$

In[601]:=
```
CenteredDiff = Series[(S[c + h] - S[c - h])/(2 h), {h, 0, 3}]
```

Out[601]= $f'[c] + \frac{1}{6} f^{(3)}[c]\, h^2 + O[h]^4$

In[602]:=
```
CenteredDiff2ndDeriv = Series[(S[c + h] - 2 S[c] + S[c - h])/h^2, {h, 0, 3}]
```

Out[602]= $f''[c] + \frac{1}{12} f^{(4)}[c]\, h^2 + O[h]^4$

Solving for Higher-order One-Sided Finite Difference Schemes

$$f^{(k)}(0) = \sum^{n}{}_{i=1} c_i\, f(x_i)$$

In[603]:=
```
nSamples = 10;
```

In[604]:=
```
iSamples = Table[i, {i, 0, nSamples - 1}]
```

Out[604]= {0, 1, 2, 3, 4, 5, 6, 7, 8, 9}

In[605]:=
```
Phi = Table[iSamples^j / Factorial[j], {j, 0, nSamples - 1}];
Phi[[1, 1]] = 1; MatrixForm[Phi]
```

Power: Indeterminate expression 0^0 encountered.

Out[605]//MatrixForm=

$$\begin{pmatrix}
1 & 1 & 1 & 1 & 1 & 1 & 1 & 1 & 1 & 1 \\
0 & 1 & 2 & 3 & 4 & 5 & 6 & 7 & 8 & 9 \\
0 & \frac{1}{2} & 2 & \frac{9}{2} & 8 & \frac{25}{2} & 18 & \frac{49}{2} & 32 & \frac{81}{2} \\
0 & \frac{1}{6} & \frac{4}{3} & \frac{9}{2} & \frac{32}{3} & \frac{125}{6} & 36 & \frac{343}{6} & \frac{256}{3} & \frac{243}{2} \\
0 & \frac{1}{24} & \frac{2}{3} & \frac{27}{8} & \frac{32}{3} & \frac{625}{24} & 54 & \frac{2401}{24} & \frac{512}{3} & \frac{2187}{8} \\
0 & \frac{1}{120} & \frac{4}{15} & \frac{81}{40} & \frac{128}{15} & \frac{625}{24} & \frac{324}{5} & \frac{16807}{120} & \frac{4096}{15} & \frac{19683}{40} \\
0 & \frac{1}{720} & \frac{4}{45} & \frac{81}{80} & \frac{256}{45} & \frac{3125}{144} & \frac{324}{5} & \frac{117649}{720} & \frac{16384}{45} & \frac{59049}{80} \\
0 & \frac{1}{5040} & \frac{8}{315} & \frac{243}{560} & \frac{1024}{315} & \frac{15625}{1008} & \frac{1944}{35} & \frac{117649}{720} & \frac{131072}{315} & \frac{531441}{560} \\
0 & \frac{1}{40320} & \frac{2}{315} & \frac{729}{4480} & \frac{512}{315} & \frac{78125}{8064} & \frac{1458}{35} & \frac{823543}{5760} & \frac{131072}{315} & \frac{4782969}{4480} \\
0 & \frac{1}{362880} & \frac{4}{2835} & \frac{243}{4480} & \frac{2048}{2835} & \frac{390625}{72576} & \frac{972}{35} & \frac{5764801}{51840} & \frac{1048576}{2835} & \frac{4782969}{4480}
\end{pmatrix}$$

In[606]:=
```
InvPhi = Inverse[Phi]; MatrixForm[InvPhi]
```

Out[606]//MatrixForm=

$$\begin{pmatrix}
1 & -\frac{7129}{2520} & \frac{6515}{1008} & -\frac{4523}{378} & \frac{285}{16} & -\frac{3013}{144} & \frac{75}{4} & -\frac{145}{12} & 5 & -1 \\
0 & 9 & -\frac{4609}{140} & \frac{42417}{560} & -\frac{7667}{60} & \frac{7807}{48} & -154 & \frac{413}{4} & -44 & 9 \\
0 & -18 & \frac{5869}{70} & -\frac{62511}{280} & \frac{24901}{60} & -\frac{6787}{12} & 563 & -392 & 172 & -36 \\
0 & 28 & -\frac{6289}{45} & \frac{72569}{180} & -\frac{4013}{5} & \frac{13873}{12} & -1203 & 868 & -392 & 84 \\
0 & -\frac{63}{2} & \frac{6499}{40} & -\frac{19557}{40} & \frac{122249}{120} & -\frac{36769}{24} & \frac{3313}{2} & -\frac{2471}{2} & 574 & -126 \\
0 & \frac{126}{5} & -\frac{265}{2} & \frac{3273}{8} & -\frac{5273}{6} & \frac{32773}{24} & -1525 & \frac{2345}{2} & -560 & 126 \\
0 & -14 & \frac{6709}{90} & -\frac{84307}{360} & \frac{10279}{20} & -\frac{9823}{12} & 939 & -742 & 364 & -84 \\
0 & \frac{36}{7} & -\frac{967}{35} & \frac{12303}{140} & -\frac{2939}{15} & \frac{3817}{12} & -373 & 302 & -152 & 36 \\
0 & -\frac{9}{8} & \frac{3407}{560} & -\frac{5469}{280} & \frac{10579}{240} & -\frac{3487}{48} & \frac{347}{4} & -\frac{287}{4} & 37 & -9 \\
0 & \frac{1}{9} & -\frac{761}{1260} & \frac{29531}{15120} & -\frac{89}{20} & \frac{1069}{144} & -9 & \frac{91}{12} & -4 & 1
\end{pmatrix}$$

In[607]:=
```
(* cHat for f^(1) *)
MatrixForm[InvPhi[[All, 2]] ]
```

Out[607]//MatrixForm=

$$\begin{pmatrix} -\frac{7129}{2520} \\ 9 \\ -18 \\ 28 \\ -\frac{63}{2} \\ \frac{126}{5} \\ -14 \\ \frac{36}{7} \\ -\frac{9}{8} \\ \frac{1}{9} \end{pmatrix}$$

In[608]:=
```
(* Define a test function *)
testFunc[x_] := Sin[x]
```

In[609]:=
```
(* Our finite-diff estimate of f^(k)(0) using h-spaced samples *)
fk[k_, h_] := (InvPhi[[All, k + 1]]).testFunc[h iSamples] / h^k
```

In[610]:=
```
fkExact[k_] := (D[testFunc[x], {x, k}]) /. {x → 0} (* Exact value of f^(k)(0) *)
```

In[611]:=
```
fkExact[1] (* First derivative at x=0 is Cos(0)=1 *)
```

Out[611]= 1

In[612]:=
```
Series[fk[1, H] - fkExact[1], {H, 0, nSamples + 1}]
```

Out[612]= $-\frac{9\,H^{10}}{22} + O[H]^{12}$

In[613]:=
```
Plot[Abs[fk[1, h] - fkExact[1]], {h, 0, 1}]
```

Out[613]=



In[614]:=
```
Plot[Log10[Abs[fk[1, h] - fkExact[1]]], {h, 0, 0.1}]
```

Out[614]=



In[615]:=
```
h0 = 1/8;
```

In[616]:=
```
(* Finite difference estimate *)
N[fk[1, h0], 60]
```

Out[616]= 0.999999999637572856620137715080028407339400569987804661470403

In[617]:=
```
(* Relative Error*)
N[Abs[(fk[1, h0] - fkExact[1]) / fkExact[1]]]
```

Out[617]= 3.62391×10^{-10}

In[618]:=
```
ListPlot[testFunc[h0 iSamples]]
```

Out[618]=



Richardson Extrapolation

In[619]:=
```
S[x_] = Normal[Series[f[x], {x, c, 14}]] (*Longer Taylor Series expansion*)
```

Out[619]=
$$f[c] + (-c+x)\, f'[c] + \frac{1}{2}(-c+x)^2 f''[c] + \frac{1}{6}(-c+x)^3 f^{(3)}[c] + \frac{1}{24}(-c+x)^4 f^{(4)}[c] + \frac{1}{120}(-c+x)^5 f^{(5)}[c] + \frac{1}{720}(-c+x)^6 f^{(6)}[c] + \frac{(-c+x)^7 f^{(7)}[c]}{5040} + \frac{(-c+x)^8 f^{(8)}[c]}{40\,320} + \frac{(-c+x)^9 f^{(9)}[c]}{362\,880} + \frac{(-c+x)^{10} f^{(10)}[c]}{3\,628\,800} + \frac{(-c+x)^{11} f^{(11)}[c]}{39\,916\,800} + \frac{(-c+x)^{12} f^{(12)}[c]}{479\,001\,600} + \frac{(-c+x)^{13} f^{(13)}[c]}{6\,227\,020\,800} + \frac{(-c+x)^{14} f^{(14)}[c]}{87\,178\,291\,200}$$

In[620]:=
$$DS[h\_] = Simplify\left[\frac{S[c+h] - S[c-h]}{2\,h}\right] \text{ (* } O\left(h^2\right) \text{ Centered-Difference Scheme *)}$$

Out[620]=
$$f'[c] + \frac{1}{6} h^2 f^{(3)}[c] + \frac{h^4 \left(51\,891\,840\, f^{(5)}[c] + 1\,235\,520\, h^2 f^{(7)}[c] + 17\,160\, h^4 f^{(9)}[c] + 156\, h^6 f^{(11)}[c] + h^8 f^{(13)}[c]\right)}{6\,227\,020\,800}$$

In[621]:=
```
Expand[DS[h]]
```

Out[621]=
$$f'[c] + \frac{1}{6} h^2 f^{(3)}[c] + \frac{1}{120} h^4 f^{(5)}[c] + \frac{h^6 f^{(7)}[c]}{5040} + \frac{h^8 f^{(9)}[c]}{362\,880} + \frac{h^{10} f^{(11)}[c]}{39\,916\,800} + \frac{h^{12} f^{(13)}[c]}{6\,227\,020\,800}$$

In[622]:=
$$Expand\left[3^2\, DS\left[\frac{h}{3}\right]\right]$$

Out[622]=
$$9\, f'[c] + \frac{1}{6} h^2 f^{(3)}[c] + \frac{h^4 f^{(5)}[c]}{1080} + \frac{h^6 f^{(7)}[c]}{408\,240} + \frac{h^8 f^{(9)}[c]}{264\,539\,520} + \frac{h^{10} f^{(11)}[c]}{261\,894\,124\,800} + \frac{h^{12} f^{(13)}[c]}{367\,699\,351\,219\,200}$$

In[623]:=
$$DS2[h\_] = Expand\left[\frac{DS[h] - 3^2\, DS\left[\frac{h}{3}\right]}{1 - 3^2}\right]$$

Out[623]=
$$f'[c] - \frac{h^4 f^{(5)}[c]}{1080} - \frac{h^6 f^{(7)}[c]}{40\,824} - \frac{13\, h^8 f^{(9)}[c]}{37\,791\,360} - \frac{41\, h^{10} f^{(11)}[c]}{13\,094\,706\,240} - \frac{671\, h^{12} f^{(13)}[c]}{33\,427\,213\,747\,200}$$

See a pattern? :) ...
Continuing Recursively...

In[624]:= $DS4[h\_] = Expand\left[\frac{DS2[h] - 3^4\, DS2\left[\frac{h}{3}\right]}{1 - 3^4}\right]$

Out[624]= $f'[c] + \frac{h^6 f^{(7)}[c]}{3\,674\,160} + \frac{13\, h^8 f^{(9)}[c]}{3\,061\,100\,160} + \frac{533\, h^{10} f^{(11)}[c]}{13\,637\,201\,212\,800} + \frac{27\,511\, h^{12} f^{(13)}[c]}{109\,657\,974\,697\,689\,600}$

In[625]:= $DS6[h\_] = Expand\left[\frac{DS4[h] - 3^6\, DS4\left[\frac{h}{3}\right]}{1 - 3^6}\right]$

Out[625]= $f'[c] - \frac{h^8 f^{(9)}[c]}{192\,849\,310\,080} - \frac{41\, h^{10} f^{(11)}[c]}{773\,229\,308\,765\,760} - \frac{27\,511\, h^{12} f^{(13)}[c]}{79\,940\,663\,554\,615\,718\,400}$

In[626]:= $DS8[h\_] = Expand\left[\frac{DS6[h] - 3^8\, DS6\left[\frac{h}{3}\right]}{1 - 3^8}\right]$

Out[626]= $f'[c] + \frac{h^{10} f^{(11)}[c]}{139\,181\,275\,577\,836\,800} + \frac{671\, h^{12} f^{(13)}[c]}{12\,950\,387\,495\,847\,746\,380\,800}$

In[627]:= $DS10[h\_] = Simplify\left[\frac{DS8[h] - 3^{10}\, DS8\left[\frac{h}{3}\right]}{1 - 3^{10}}\right]$ (* An $O(h^{12})$ finite difference approx! *)

Out[627]= $f'[c] - \frac{h^{12} f^{(13)}[c]}{1\,282\,088\,362\,088\,926\,891\,699\,200}$

Romberg Integration

In[628]:= $g[x\_] := 4\, Sqrt\left[1 - x^2\right]$

In[629]:= `Integrate[g[x], {x, 0, 1}]`

Out[629]= π

In[630]:= `NIntegrate[g[x], {x, 0, 1}, 12]`

NIntegrate: Invalid integration variable or limit(s) in 12.

Out[630]= `NIntegrate[g[x], {x, 0, 1}, 12]`

In[631]:= `NIntegrate[g[x], {x, 0, 1}, WorkingPrecision → 100]`

Out[631]= 3.14159265358979323846264338327950288419716939937510582097494459230781640628620899862803482534211706 8

In[632]:=